

SimpleTensor - a user-friendly Mathematica package for elementary tensor and differential-geometric calculations

D. O. Rybalka¹

¹*Max Planck Institute for the Physics of Complex Systems, Nöthnitzer Str. 38, 01187 Dresden, Germany*

(Dated: January 14, 2022)

In this paper we present a short overview of the new Wolfram Mathematica package intended for elementary "in-basis" tensor and differential-geometric calculations. In contrast to alternatives our package is designed to be easy-to-use, short, all-purpose, and hackable. It supports tensor contractions using Einstein notation, transformations between different bases, tensor derivative operator, expansion in basis vectors and forms, exterior derivative, and interior product.

I. INTRODUCTION

As a motivation for this project can serve a simple expression from electrodynamics: the dual electromagnetic tensor $\frac{1}{2}\epsilon^{\mu\nu\alpha\beta}\partial_\alpha A_\beta$. To the best of our knowledge there is no easy way to calculate this expression in base Mathematica without shuffling explicitly index contractions and paying close attention to metric signatures. Furthermore, the search for an external package proved to be unsatisfactory as well, with the results being hard to use and containing large complexity overhead for such a simple task. Therefore, we created a user-friendly tensor computation package for situations, where clarity is more important than performance.

After 12 versions and 33 years of development Wolfram Mathematica is an invaluable tool for both scientists and students. Its functionality covers most needs in mathematical calculations starting from undergraduate courses up to physics-theoretical research. The most glaring exceptions in its scope is the tensor calculus (which is present only in the form of explicit lists manipulation) and differential geometry.

The reason for that may be the fact that these topics cannot be fitted well in Mathematica's paradigm of "stateless" computations, i.e., the idea that each function must use only the arguments explicitly provided to it. And while this approach works well in most cases, in tensor calculus and differential geometry it becomes burdensome. For example, a simple contraction of tensors require not only the actual tensor components, but also dimensions and often metrics of spaces where the corresponding indices are defined.

There already exist a few packages intended to provide this functionality, for example, `xAct` [1], `TensoriaCalc` [2], `grt` [3], `Ricci` [4] just to name a few. However, in our opinion, they all suffer from one or more of the following flaws:

- The package is old, outdated, or not supported anymore;
- The package is hard to master (or even to grasp), as it contains dozens or even hundreds of public functions;
- The package is focused primarily on the General Relativity computations;
- The package is hard to adapt for your personal needs.

With this paper we introduce SimpleTensor - a yet-another package for elementary tensor and differential-geometric calculations, intended to fill the gap in base Mathematica functionality. As the name suggests, it differs from the alternatives by being accessible, concise (only 11 public functions), up-to-date, all-purpose, and well-documented (hackable). It also provides basis for elementary in-basis differential-geometric computations.

However, some sacrifices had to be made in order to achieve these goals. SimpleTensor is not optimized for any large-scale computations and will be slower than the alternatives. It was also not specialized for any particular area of physics or mathematics and so loses considerably on functionality for specialized computations. Nevertheless, even despite these flaws we hope that SimpleTensor can find its auditory among both students and scientists.

The latest version of SimpleTensor can be found in the author's GitHub repository [5]. Located there are the source code, the generated `.m` package file, together with the tests and the example session files. The tests file can be used to get a quick overview of all the capabilities of SimpleTensor as well as to test custom modifications of the source code (enhances hackability of the package). The example session will be examined in detail later in this paper.

In the next chapter we present the technical information concerning SimpleTensor, discuss the package design and document provided public functions.

II. TECHNICAL INFORMATION

The package installation process is similar to the most other external Mathematica packages. One may simply copy the `SimpleTensor.m` file to the Applications folder in the `$UserBaseDirectory`. The package then can be loaded as any other Mathematica package with `Needs` command:

```
In[1]:= Needs["SimpleTensor`"]
```

Alternatively, if any source code modifications are desirable, one may simply edit and execute the contents of `SimpleTensor.nb` file.

Let us now give a broad overview of the package. The main actors of `SimpleTensor` are tensors, which have to be explicitly defined by their components in some coordinate system in some space. Spaces are specified by a unique name, dimension, indices, and coordinates. Both indices and coordinates are the usual Mathematica symbols, that have to be unique among all spaces to avoid name collisions. Indices can be either contravariant (upper) - denoted with a positive sign, e.g., $+\mu$ or simply μ , or covariant (lower) - denoted with a negative sign, e.g., $-\mu$. In tensors the indices are then denoted in square brackets, so that F^μ_ν becomes `F[μ , - ν]`.

Tensor components can be assigned using the `SetTensor` command (\equiv operator). The main command of `SimpleTensor` is `GetArray`, which performs all tensor calculations on an expression and returns the resulting array. Special meaning is reserved for a few tensor names: `metric` is used to lower and raise indices as well as for basis change, `basis` is a tensor of differential-geometric basis forms/vectors, and `D` is the tensor derivative. Basic differential geometric operations are provided by `DirectProduct` (\otimes), `WedgeProduct` (\wedge), `ExteriorDerivative`, and `InteriorProduct`. Below we document the public methods of `SimpleTensor` in more detail.

`DefineSpace[name, dim]`: defines a space of dimension `dim` with a string `name` on which tensors live. The optional variables are:

- `index` - a list of indices on this space or a string prefix to generate names automatically; if missing, name is used as a prefix.
- `coord` - a list of coordinates on this space or a string prefix to generate names automatically; if missing, name is used as a prefix.

For example, we can define a 4-dimensional Minkowski space with `DefineSpace["Minkowski", 4, index \rightarrow { μ , ν , a , β }, coord \rightarrow { t , x , y , z }]`. This will additionally define a set of basis vectors $\partial_t, \partial_x, \partial_y, \partial_z$ and a set of basis forms `dt, dx, dy, dz`.

If `index` or `coord` optional arguments are given as a string then it is used as a prefix. For example, `index \rightarrow "L"` is equivalent to `index \rightarrow {iL0, iL1, ...}` and `coord \rightarrow "L"` to `coord \rightarrow {xL0, xL1, ...}`. If `index` or `coord` are not specified then the space `name` is used as a prefix.

Defining a space for a second time deletes all definitions connected to the old one. Thus, one can effectively delete a space by redefining it with zero dimension.

`SetTensor[t[μ], arr]`: associates array `arr` to a tensor `t[μ]`. For convenience this function overrides the built-in `Congruent` (\equiv) function. To change the default behavior edit corresponding lines in the package file.

Naturally, any tensor name (head) and upper/lower indices can be used here. Indices from the same space can be used interchangeably. For example, in the Minkowski space defined above `t[μ , - ν]`, `t[ν , - μ]`, `t[a , - β]`, and even `t[μ , - μ]` inside `SetTensor` mean the same.

The dimension compatibility of `arr` and indices must be correct and will be checked on assignment. It is possible, however, to have extra dimensions in `arr`, in which case tensor elements are assumed to be matrices. We will demonstrate in the next chapter how this allows to define a tensor of Dirac matrices γ^μ .

Note, that assignments for the same tensor head but different signatures or spaces are independent and do not overwrite each other. This allows for simpler expressions (especially if `metric` is complicated), but the burden of ensuring consistency in this case falls on user's shoulders.

`GetArray[expr]`: substitutes tensor variables with arrays and performs contractions and math operations. The optional variables are:

- `indices` - a list of indices in which order the tensor should be transposed. If omitted `indices` are taken to be the free indices in `expr` in standard order, i.e., the one returned by the standard `Sort[]` function.

Be mindful of the standard ordering as `GetArray[F[μ , ν]] \neq GetArray[F[ν , μ]] in general. For example, for antisymmetric tensors GetArray[F[μ , ν] + F[ν , μ]] == 0 as expected. However, they are equal if one specifies the desired indices explicitly GetArray[F[μ , ν]] == GetArray[F[μ , ν], { μ , ν }] == GetArray[F[ν , μ], { ν , μ }]`

metric[-i1, -i2]: is the metric tensor for the space in which **i1** and **i2** are defined. **metric[i1, i2]** is the inverse metric tensor. If **i1** and **i2** belong to different spaces then it works as a tetrad (vielbein) tensor and represents the change of basis.

The **metric** tensor is automatically used by **GetArray** function to transform indices between different signatures and spaces. Multiple metrics will be chained where needed for complex space-signature transformations.

The inverse **metric** will be automatically calculated on the assignment unless it was already explicitly assigned earlier. This allows for simpler expressions for complicated metrics, but consistency needs to be insured by the user.

basis[i]: is the tensor of basis forms for the space in which index **i** is defined. **basis[-i]** is the tensor of basis vectors.

The **basis** tensor is protected and may not be modified by the user. Its value is automatically generated from coordinates during the **DefineSpace** call. Note, that **basis[i]** and **basis[-i]** are essentially different and are not connected through **metric**.

∂[±i]: is the tensor derivative for the space in which index **i** is defined. **∂space[±i]** is also defined for a derivative in the space **space**, which can be used with any **i**.

In the Minkowski space example the components of $\partial[-\mu]$ are $(\partial_t, \partial_x, \partial_y, \partial_z)$. The argument of the derivative must be given in square brackets after the tensor $\partial[-\mu][arg]$ and may include other tensors. Note, that $\partial[-\mu] \neq \text{metric}[-\mu, a] \partial[-a]$ if μ and **a** are from different spaces. However, the space-specific derivatives satisfy this equality for any indices.

Tensor names starting with ∂ are reserved for tensor derivatives and may not be modified. Their values are automatically generated from coordinates during the **DefineSpace** call.

TensorPlus[args]: is used to add matrix-valued tensors (for example, the set of Dirac matrices γ^μ) with explicit lists (for example, **IdentityMatrix[4]**). It is equivalent to **Plus[]** in most cases, except it prevents the automatic threading over lists. For convenience this function overrides the built-in **CirclePlus** (\oplus) function.

DirectProduct[args]: represents the tensor product of basis vectors and forms. It expands **args** automatically until a sum of basis tensor products is achieved. For convenience this function overrides the built-in **CircleTimes** (\otimes) function.

WedgeProduct[args]: represents the wedge product of basis forms. It expands **args** automatically until a sum of basis wedge products is achieved. For convenience this function overrides the built-in **Wedge** (\wedge) function.

The **WedgeProduct** of basis forms is automatically sorted into canonical order respecting anticommutation rules. If a basis form is repeated inside **WedgeProduct** it is set to zero. Both **WedgeProduct** and **DirectProduct** also work with the **basis** tensor.

ExteriorDerivative[space][expr]: gives the exterior derivative of **expr** in **space**. If latter is a **Sequence** of spaces, the exterior derivative is taken in their product space. If **space** is empty, the exterior derivative is taken in the product space of all defined spaces.

InteriorProduct[expr1, expr2]: is the symmetric interior product of basis vectors and forms in **expr1** and **expr2**. The **DirectProduct**'s of bases are multiplied successively starting from the first position.

Basis vectors and forms correspond to each other with respect to the **InteriorProduct**, so that for the Minkowski space example **InteriorProduct[dx, ∂x] == 1** and **GetArray@InteriorProduct[basis[-μ], basis[ν]] == IdentityMatrix[4]**. The interior product of composed expressions is performed successively starting from the first position, so that **InteriorProduct[∂y ⊗ ∂x, dx ∧ dy] == - InteriorProduct[∂x, dx] = -1**.

As one can see from the documentation above there are only 11 public functions in the whole package. Moreover, if one is not interested in differential geometry then the last 4 functions are irrelevant and can be simply ignored. In the next chapter we present a few illustrative examples of how this limited functionality can already be used for various physics computations. The notebook session with these examples, as well as the test-file with edge cases can be found in the project repository [5].

III. EXAMPLES

A. Electromagnetic tensor

In the first part we examine the applications of SimpleTensor on the example of undergraduate-level relativistic electromagnetism. The following material can be found in most textbooks on classical electrodynamics [6, 7]. In this section we will first derive the relations between electromagnetic potentials and fields. We will then calculate the field part of Maxwell equations in 2 different ways: explicitly from the EM tensor and using exterior calculus from the potential 1-form. Finally, we will obtain the field transformations under a boost using a moving reference frame.

We start by defining a Minkowski space with mostly negative metric.

```
In[2]:= DefineSpace["Minkowski", 4, index -> {μ, ν, λ, ρ}, coord -> {t, x, y, z}]
metric[-μ, -ν] ≡ DiagonalMatrix[{1, -1, -1, -1}]
```

Next we set a Levi-Civita tensor $\epsilon^{\mu\nu\lambda\rho}$ and a time direction vector T^μ . The latter is done using the full function name as illustration.

```
In[4]:= ε[μ, ν, λ, ρ] ≡ Normal@LeviCivitaTensor[4]
SetTensor[T[μ], {1, 0, 0, 0}]
```

Finally, we set a general electromagnetic 4-potential A^μ . Note, that we use the same coordinates defined in the Minkowski space.

```
In[6]:= A[μ] ≡ {φ[t, x, y, z], A1[t, x, y, z], A2[t, x, y, z], A3[t, x, y, z]}
```

Individual components of a tensor can be extracted by contracting it with a corresponding unit vector. For example, by contracting EM 4-potential A^μ with a time direction vector T^μ we get the corresponding zeroth component - the scalar potential ϕ . Alternatively one can first calculate the array and then access its components.

```
In[7]:= A[-μ] T[μ] // GetArray
First@GetArray[A[-μ]] == %
```

```
Out[7]= φ[t, x, y, z]
```

```
Out[8]= True
```

Using the partial derivative tensor operator ∂_μ we can calculate the electromagnetic tensor $F_{\mu\nu} = \partial_\mu A_\nu - \partial_\nu A_\mu$ and assign it to a tensor.

```
In[9]:= ∂[-μ][A[-ν]] - ∂[-ν][A[-μ]] // GetArray;
F[-μ, -ν] ≡ %
```

The antisymmetry of EM tensor can be shown either elementwise or by summing the arrays of normal and transposed arrays.

```
In[11]:= GetArray[F[-μ, -ν] + F[-ν, -μ]]
GetArray[F[-μ, -ν], {-μ, -ν}] + GetArray[F[-μ, -ν], {-ν, -μ}]
```

```
Out[11]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

```
Out[12]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

From the electromagnetic tensor we can extract the vectors of the electric $E^\mu = F^{\mu\nu}T_\nu = (0, -\vec{\nabla}\phi - \partial_t\vec{A})$ and the magnetic fields $B^\mu = \frac{1}{2}\epsilon^{\mu\nu\lambda\rho}T_\nu F_{\lambda\rho} = (0, \vec{\nabla}\times\vec{A})$ (both purely spatial).

```
In[13]:= F[μ, ν] T[-ν] // GetArray // Column
1/2 ε[μ, ν, λ, ρ] F[-λ, -ρ] T[-ν] // GetArray // Column
```

```
Out[13]= 0
-φ(0,1,0,0)[t, x, y, z] - A1(1,0,0,0)[t, x, y, z]
-φ(0,0,1,0)[t, x, y, z] - A2(1,0,0,0)[t, x, y, z]
-φ(0,0,0,1)[t, x, y, z] - A3(1,0,0,0)[t, x, y, z]
```

```

Out[14]= 0
A3(0,0,1,0)[t,x,y,z]-A2(0,0,0,1)[t,x,y,z]
A1(0,0,0,1)[t,x,y,z]-A3(0,1,0,0)[t,x,y,z]
A2(0,1,0,0)[t,x,y,z]-A1(0,0,1,0)[t,x,y,z]

```

Let us now derive the field part of Maxwell equations. First we define a general electromagnetic tensor using the electric and magnetic field components.

```

In[15]:= Ffield[-μ,-ν] ≡
{{ 0, Ex[t,x,y,z], Ey[t,x,y,z], Ez[t,x,y,z]},
{-Ex[t,x,y,z], 0, -Bz[t,x,y,z], By[t,x,y,z]},
{-Ey[t,x,y,z], Bz[t,x,y,z], 0, -Bx[t,x,y,z]},
{-Ez[t,x,y,z], -By[t,x,y,z], Bx[t,x,y,z], 0}}

```

Next we calculate the field part of the homogeneous Maxwell equations $\partial_\lambda F_{\mu\nu}$ antisymmetrized by all indices. For illustration we filter out trivial components. This gives the well-known $\vec{\nabla} \times \vec{E} + \partial_t \vec{B} = 0$ and $\vec{\nabla} \cdot \vec{B} = 0$.

```

In[16]:= D[-λ][Ffield[-μ,-ν]] + D[-μ][Ffield[-ν,-λ]] + D[-ν][Ffield[-λ,-μ]]
// GetArray // Flatten // DeleteCases[0] // DeleteDuplicatesBy[Sort[{-#,#}]&] // Column
Out[16]= Ex(0,0,1,0)[t,x,y,z]-Ey(0,1,0,0)[t,x,y,z]-Bz(1,0,0,0)[t,x,y,z]
Ex(0,0,0,1)[t,x,y,z]-Ez(0,1,0,0)[t,x,y,z]+By(1,0,0,0)[t,x,y,z]
Ey(0,0,0,1)[t,x,y,z]-Ez(0,0,1,0)[t,x,y,z]-Bx(1,0,0,0)[t,x,y,z]
-Bz(0,0,0,1)[t,x,y,z]-By(0,0,1,0)[t,x,y,z]-Bx(0,1,0,0)[t,x,y,z]

```

Finally, we calculate the field part of the inhomogeneous Maxwell equations $\partial_\mu F^{\mu\nu}$. This gives the correct $\vec{\nabla} \cdot \vec{E}$ and $\vec{\nabla} \times \vec{B} - \partial_t \vec{E}$.

```

In[17]:= D[-μ][Ffield[μ,ν]] // GetArray // Column
Out[17]= Ez(0,0,0,1)[t,x,y,z]+Ey(0,0,1,0)[t,x,y,z]+Ex(0,1,0,0)[t,x,y,z]
-By(0,0,0,1)[t,x,y,z]+Bz(0,0,1,0)[t,x,y,z]-Ex(1,0,0,0)[t,x,y,z]
Bx(0,0,0,1)[t,x,y,z]-Bz(0,1,0,0)[t,x,y,z]-Ey(1,0,0,0)[t,x,y,z]
-Bx(0,0,1,0)[t,x,y,z]+By(0,1,0,0)[t,x,y,z]-Ez(1,0,0,0)[t,x,y,z]

```

The above calculation can also be done using the exterior calculus. Let us obtain the electromagnetic potential 1-form A by contracting the potential A_μ with the basis of 1-forms dx^μ .

```

In[18]:= Aform = A[-μ] basis[μ] // GetArray
Out[18]= dt φ[t,x,y,z] - dx A1[t,x,y,z] - dy A2[t,x,y,z] - dz A3[t,x,y,z]

```

The electromagnetic field 2-form is the exterior derivative of the potential 1-form $F = dA$. It is equal to the previously calculated electromagnetic tensor. Note, that in our simple case we do not need to specify the space in which exterior derivative is taken.

```

In[19]:= Fform = ExteriorDerivative[] [Aform] // FullSimplify
Fform == F[-μ,-ν] * 1/2 basis[μ] ^ basis[ν] // GetArray // FullSimplify
Out[19]= dy∧dz (A2(0,0,0,1)[t,x,y,z] - A3(0,0,1,0)[t,x,y,z])
+dx∧dy (A1(0,0,1,0)[t,x,y,z] - A2(0,1,0,0)[t,x,y,z])
+dx∧dz (A1(0,0,0,1)[t,x,y,z] - A3(0,1,0,0)[t,x,y,z])
-dt∧dx (φ(0,1,0,0)[t,x,y,z] + A1(1,0,0,0)[t,x,y,z])
-dt∧dy (φ(0,0,1,0)[t,x,y,z] + A2(1,0,0,0)[t,x,y,z])
-dt∧dz (φ(0,0,0,1)[t,x,y,z] + A3(1,0,0,0)[t,x,y,z])
Out[20]= True

```

The homogeneous Maxwell equations are trivially satisfied $dF = ddA = 0$.

```

In[21]:= ExteriorDerivative[] [Fform] // FullSimplify

```

Out[21]= 0

The field part of the inhomogeneous Maxwell equations $d(\star F)$ (for potentials) can be derived using an improvised Hodge star operator. Although the result is cumbersome it is not hard to see that it reproduces the correct expression $\square A^\mu - \partial^\mu(\partial_\nu A^\nu)$.

```
In[22]:= HodgeStar[f_] := Module[{tmp},
  tmp = f /. {a_&wedge;b_ -> a&wedge;b - b&wedge;a};
  tmp = tmp /. {dt -> dtilde t, dx -> dtilde x, dy -> dtilde y, dz -> dtilde z};
  1/2 InteriorProduct[tmp, dt&wedge;dx&wedge;dy&wedge;dz]
];
ExteriorDerivative[] [HodgeStar[Fform]] // FullSimplify

Out[23]= -dx&wedge;dy&wedge;dz (phi^(0,2,0,0)[t,x,y,z] + phi^(0,0,2,0)[t,x,y,z] + phi^(0,0,0,2)[t,x,y,z]
+ A1^(1,1,0,0)[t,x,y,z] + A2^(1,0,1,0)[t,x,y,z] + A3^(1,0,0,1)[t,x,y,z])
-dt&wedge;dy&wedge;dz (A1^(2,0,0,0)[t,x,y,z] + A1^(0,0,2,0)[t,x,y,z] + A1^(0,0,0,2)[t,x,y,z]
+ phi^(1,1,0,0)[t,x,y,z] - A2^(0,1,1,0)[t,x,y,z] - A3^(0,1,0,1)[t,x,y,z])
+dt&wedge;dx&wedge;dz (A2^(2,0,0,0)[t,x,y,z] + A2^(0,2,0,0)[t,x,y,z] + A2^(0,0,2,0)[t,x,y,z]
+ phi^(1,0,1,0)[t,x,y,z] - A1^(0,1,1,0)[t,x,y,z] - A3^(0,0,1,1)[t,x,y,z])
-dt&wedge;dx&wedge;dy (A3^(2,0,0,0)[t,x,y,z] + A3^(0,2,0,0)[t,x,y,z] + A3^(0,0,2,0)[t,x,y,z]
+ phi^(1,0,0,1)[t,x,y,z] - A1^(0,1,0,1)[t,x,y,z] - A2^(0,0,1,1)[t,x,y,z])
```

Lastly, let us derive the field transformations under a boost. First we define a general 4-dimensional space. The indices of this space are `iMoving0`, `iMoving1`, etc..

```
In[24]:= DefineSpace["Moving", 4]
```

Next we define the transformation matrix from the original space to the moving space with the help of a vielbein (tetrad) $e_\mu^i = \partial x'^i / \partial x^\mu$, where $x'^i = x'^i(x)$ is the coordinates of the moving frame. Let the new system be moving in the z -direction with velocity v in natural units.

```
In[25]:= metric[-mu, iMoving1] ==
  {{ 1/sqrt[1-v^2], 0, 0, -v/sqrt[1-v^2] },
   { 0, 1, 0, 0 },
   { 0, 0, 1, 0 },
   { -v/sqrt[1-v^2], 0, 0, 1/sqrt[1-v^2] }}
```

It is instructive to check that the moving space is also Minkowski. Note, that this metric was not set manually, but calculated using the tetrad $g_{ij} = g_{\mu\nu} e_i^\mu e_j^\nu$.

```
In[26]:= metric[-iMoving1, -iMoving2] // GetArray // Simplify // MatrixForm
```

```
Out[26]= ( 1 0 0 0
           0 -1 0 0
           0 0 -1 0
           0 0 0 -1 )
```

The time direction vector in the moving space is:

```
In[27]:= Tmoving[iMoving0] == {1,0,0,0}
```

Now we can easily derive the well-known transformations for electric and magnetic fields in the moving frame using the same formulae as before.

```
In[28]:= Ffield[iMoving0, iMoving1] Tmoving[-iMoving1] // GetArray // FullSimplify
1/2 e[iMoving0, iMoving1, iMoving2, iMoving3] Tmoving[-iMoving1] Ffield[-iMoving2, -iMoving3]
// GetArray // FullSimplify
```

$$\begin{aligned} \text{Out[28]} &= \left\{ 0, \frac{\text{Ex}[t, x, y, z] - v \text{By}[t, x, y, z]}{\sqrt{1-v^2}}, \frac{\text{Ey}[t, x, y, z] + v \text{Bx}[t, x, y, z]}{\sqrt{1-v^2}}, \text{Ez}[t, x, y, z] \right\} \\ \text{Out[29]} &= \left\{ 0, \frac{\text{Bx}[t, x, y, z] + v \text{Ey}[t, x, y, z]}{\sqrt{1-v^2}}, \frac{\text{By}[t, x, y, z] - v \text{Ex}[t, x, y, z]}{\sqrt{1-v^2}}, \text{Bz}[t, x, y, z] \right\} \end{aligned}$$

As a side note we mention, that ∂ is the partial derivative in the coordinates of its index, therefore the right-hand side of the first comparison below is in fact zero. In expressions with basis changes, like $\partial_\mu F^{\mu\nu} = e_\mu^a \partial_a F^{\mu\nu} = \partial_a F^{a\mu}$, one must use ∂space to specify the coordinates explicitly, independent of the index.

```
In[30]:= FullSimplify@GetArray[ $\partial[-\mu]$  [Ffield[ $\mu, \nu$ ]]]
      === FullSimplify@GetArray[metric[iMoving0,  $-\mu$ ]  $\partial[-iMoving0]$  [Ffield[ $\mu, \nu$ ]]]
FullSimplify@GetArray[ $\partial[-\mu]$  [Ffield[ $\mu, \nu$ ]]]
      === FullSimplify@GetArray[metric[iMoving0,  $-\mu$ ]  $\partial\text{Minkowski}[-iMoving0]$  [Ffield[ $\mu, \nu$ ]]]
      === FullSimplify@GetArray[ $\partial\text{Minkowski}[-iMoving0]$  [Ffield[iMoving0,  $\nu$ ]]]

Out[30]= False
Out[31]= True
```

Finally, when the space is no longer needed the following command effectively deletes it and all definitions connected to it.

```
In[32]:= DefineSpace["Moving", 0]
DefineSpace["Minkowski", 0]
```

B. Spinors

In this part we will examine applications of SimpleTensor to Dirac and Weyl particles. The following material can be found in the first few chapters of [8], although similar treatments are also given in many other quantum physics textbooks. First, we will define the 4-vector of Dirac matrices and show how it can be manipulated into other matrices of the Clifford algebra using SimpleTensor. We will solve the Dirac and Weyl equations, find the plane-wave solutions for both spinors and calculate the corresponding Berry phases [9]. Lastly, we will check the transformations of Dirac spinors under a Lorentz boost.

We start again by defining a Minkowski space with mostly negative metric, the Levi-Civita tensor, and a time-direction 4-vector.

```
In[33]:= DefineSpace["Minkowski", 4, index  $\rightarrow$  { $\mu, \nu, \lambda, \rho, a, \beta, \gamma, \delta$ }, coord  $\rightarrow$  {t, x, y, z}]
metric[ $-\mu, -\nu$ ]  $\equiv$  DiagonalMatrix[{1, -1, -1, -1}]
 $\epsilon$ [ $\mu, \nu, \lambda, \rho$ ]  $\equiv$  Normal@LeviCivitaTensor[4]
T[ $\mu$ ]  $\equiv$  {1, 0, 0, 0}
```

Next we define a 4-vector of gamma (Dirac) matrices γ^μ in chiral representation. Note, that even though the list on the right-hand side has dimensions $\{4, 4, 4\}$ it is possible to assign it to a (matrix-valued) 4-vector γ^μ because the top-most dimension coincide with the Minkowski space dimension.

$$\text{In[34]}:= \gamma[\mu] \equiv \left\{ \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & -1 & 0 & 0 \\ -1 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & -i \\ 0 & 0 & i & 0 \\ 0 & i & 0 & 0 \\ -i & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \\ -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{pmatrix} \right\}$$

It is easy to check, for example, that the trace of Gamma matrices product is proportional to the metric $\text{tr}(\gamma^\mu \gamma^\nu) = 4g^{\mu\nu}$. Note, that the GetArray command takes care of the dot product between the matrices as well as the trace operator.

```
In[36]:= Tr[ $\gamma[\mu] \cdot \gamma[\nu]$ ] == 4metric[ $\mu, \nu$ ] // GetArray // FullSimplify

Out[36]= True
```

Using the 4-vector of gamma matrices it is possible to construct also the "fifth" gamma matrix γ_5 and a set of sigma matrices $\sigma^{\mu\nu}$. Note, that in the chiral representation γ_5 is indeed diagonal and $P_{R/L} = (1 \pm \gamma_5)/2$ are the projection operators onto the left- and right-handed components.

```
In[37]:=  $\gamma_5 = \text{GetArray}\left[\frac{-i}{24} \in [\mu, \nu, \rho, \lambda] \gamma[-\mu] \cdot \gamma[-\nu] \cdot \gamma[-\rho] \cdot \gamma[-\lambda]\right]; \gamma_5 // \text{MatrixForm}$ 
 $\sigma[\mu, \nu] \equiv \text{GetArray}\left[\frac{i}{2} (\gamma[\mu] \cdot \gamma[\nu] - \gamma[\nu] \cdot \gamma[\mu])\right];$ 
```

```
Out[37]= 
$$\begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```

Let us also introduce for the following discussion a convenient conjugate-transpose (dagger) and an adjoint (bar) operations on spinors. Note how we used $\gamma_0 = T_\mu \gamma^\mu$.

```
In[41]:= dagger[a_] := Module[{res},
  res = a /. Complex[r_, i_] -> Complex[r, -i];
  If[MatchQ[Dimensions[res], {x_, x_}], Transpose@res, res]
]
bar[psi_] := dagger[psi].GetArray[T[mu] gamma[-mu]]
```

Next let us find a plane-wave solution ψ to the Dirac equation for a particle with mass m and 4-momentum p^μ . First we define the 4-momentum and write the matrix part of the Dirac equation $(p_\mu \gamma^\mu - m)\psi = 0$. Note that when adding a list-valued tensor to an explicit Mathematica list one should use the `TensorPlus` (\oplus) operator. Otherwise Mathematica automatically threads the tensor inside the explicit list before any computation can take place yielding a list of lists in the result.

```
In[42]:= p[mu] := {p0, p1, p2, p3}
EqDirac = p[-mu] gamma[mu]  $\oplus$  -m IdentityMatrix[4] // GetArray;
EqDirac // MatrixForm
```

```
Out[42]= 
$$\begin{pmatrix} -m & 0 & p_0 - p_3 & -p_1 + i p_2 \\ 0 & -m & -p_1 - i p_2 & p_0 + p_3 \\ p_0 + p_3 & p_1 - i p_2 & -m & 0 \\ p_1 + i p_2 & p_0 - p_3 & 0 & -m \end{pmatrix}$$

```

Let us now find the dispersion. A non-trivial solution exists only when the determinant is non-zero.

```
In[43]:= Solve[Det[EqDirac] == 0, p0]
```

```
Out[43]= {{p0 -> -sqrt[m^2+p1^2+p2^2+p3^2]}, {p0 -> -sqrt[m^2+p1^2+p2^2+p3^2]},
  {p0 -> sqrt[m^2+p1^2+p2^2+p3^2]}, {p0 -> sqrt[m^2+p1^2+p2^2+p3^2]}}
```

This is a doubly-degenerate dispersion for Dirac particles and anti-particles. Let us choose the positive branch and find a conveniently normalized bi-spinor solution.

```
In[44]:= EqDirac /. p0 -> sqrt[m^2+p1^2+p2^2+p3^2] // NullSpace
psi =  $\frac{\text{First}[\%]}{\sqrt{p_3 + \sqrt{m^2+p_1^2+p_2^2+p_3^2}}}$ 
```

```
Out[44]= {{- $\frac{p_1 - i p_2}{m}$ ,  $\frac{p_3 + \sqrt{m^2+p_1^2+p_2^2+p_3^2}}{m}$ , 0, 1}, {- $\frac{p_3 + \sqrt{m^2+p_1^2+p_2^2+p_3^2}}{m}$ ,  $-\frac{p_1 + i p_2}{m}$ , 1, 0}}
```

We check that this spinor solution indeed has the current $j^\mu = \bar{\psi} \gamma^\mu \psi = 2p^\mu$ (factor 2 due to ψ being a bi-spinor).

```
In[46]:= bar[psi].gamma[mu].psi // GetArray // FullSimplify
```

```
Out[46]= {2 sqrt[m^2+p1^2+p2^2+p3^2], 2 p1, 2 p2, 2 p3}
```


As a more advanced result, the Berry connection $\vec{a}_p = -i\bar{\psi}\vec{\nabla}_p\psi$ of the bi-spinor is zero because the Berry connections of comprising left and right spinors cancel each other $\vec{a}_p^R = -\vec{a}_p^L$.

```
In[47]:= -i bar[\psi].D[\psi, #] & /@ {p1, p2, p3} // FullSimplify
          \psi_L = \frac{1}{2} (IdentityMatrix[4] - \gamma_5) . \psi; \psi_R = \frac{1}{2} (IdentityMatrix[4] + \gamma_5) . \psi;
          bar[\psi_R].D[\psi_L, #] == - bar[\psi_L].D[\psi_R, #] & /@ {p1, p2, p3} // FullSimplify

Out[47]= {0, 0, 0}

Out[50]= {True, True, True}
```

Let us now quickly calculate the Berry potential for a Weyl particle, as it will be needed in the next part. The calculation is very similar to the one above. First we define a 4-vector of Pauli matrices σ^μ and write the equation for a right-handed Weyl spinor $p_\mu\sigma^\mu\psi = 0$.

```
In[51]:= \sigma[\mu] \equiv Table[PauliMatrix[i], {i, 0, 3}]
          EqWeyl = p[-\mu] \sigma[\mu] // GetArray;
          EqWeyl // MatrixForm

Out[51]= \left( \begin{array}{cc} p_0 - p_3 & -p_1 + i p_2 \\ -p_1 - i p_2 & p_0 + p_3 \end{array} \right)
```

Next we find the dispersion.

```
In[52]:= Solve[Det[EqWeyl] == 0, p0]

Out[52]= {{p0 \to -\sqrt{p1^2 + p2^2 + p3^2}}, {p0 \to \sqrt{p1^2 + p2^2 + p3^2}}}
```

Now we find the right-handed Weyl spinor solution u_R and normalize it to 1.

```
In[53]:= EqWeyl /. p0 \to \sqrt{p1^2 + p2^2 + p3^2} // NullSpace
          uR = FullSimplify[Normalize@First[%], {p1 \in Reals, p2 \in Reals, p3 \in Reals}];

Out[53]= {{\frac{p3 + \sqrt{p1^2 + p2^2 + p3^2}}{p1 + i p2}, 1}}
```

Finally we calculate the Berry potential $\vec{a}_p = -i\bar{u}_R\vec{\nabla}_p u_R$ and check that it indeed gives the monopole in momentum space at the coordinate origin $\vec{\nabla} \times \vec{a}_p = \frac{1}{2}\vec{p}/|p|^3$.

```
In[55]:= -i dagger[uR].D[uR, #] & /@ {p1, p2, p3} // FullSimplify
          Curl[%, {p1, p2, p3}] // FullSimplify

Out[55]= {\frac{p2 + \frac{p2 p3}{\sqrt{p1^2 + p2^2 + p3^2}}}{2(p1^2 + p2^2)}, -\frac{p1 + \frac{p1 p3}{\sqrt{p1^2 + p2^2 + p3^2}}}{2(p1^2 + p2^2)}, 0}

Out[56]= {\frac{p1}{2(p1^2 + p2^2 + p3^2)^{3/2}}, \frac{p2}{2(p1^2 + p2^2 + p3^2)^{3/2}}, \frac{p3}{2(p1^2 + p2^2 + p3^2)^{3/2}}}
```

Let us now check that the Dirac spinor transforms correctly under Lorentz transformations. For simplicity we consider a Lorentz boost in z-direction and parametrize it with an anti-symmetric tensor $w_{\mu\nu}$. As we will check later v is the speed of the moving frame.

```
In[57]:= w[-\mu, -\nu] \equiv \left( \begin{array}{ccc} 0 & 0 & 0 \text{ Log}\left[\frac{1+v}{\sqrt{1-v^2}}\right] \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ -\text{Log}\left[\frac{1+v}{\sqrt{1-v^2}}\right] & 0 & 0 \end{array} \right)
```

Lorentz 4-vectors transform using the following representation Λ_{vec} of the Lorentz group, where $J_{\alpha\beta}^{\mu\nu}$ is the transformation generator (see [8] for details). Note, that p_{new} is indeed the momentum boosted with speed v in z-direction.

```
In[58]:= J[μ, ν, -a, -β] ≡ Table[i (δμ,a δν,β - δμ,β δν,a), {μ, 0, 3}, {ν, 0, 3}, {a, 0, 3}, {β, 0, 3}]
Avec = MatrixExp @ GetArray[- $\frac{i}{2}$  w[-μ, -ν] J[μ, ν, a, -β]] // FullSimplify;
pnew = Avec.GetArray[p[μ]] // FullSimplify

Out[60]= { $\frac{p_0 + p_3 v}{\sqrt{1-v^2}}$ , p1, p2,  $\frac{p_3 + p_0 v}{\sqrt{1-v^2}}$ }
```

Dirac spinors transform under Λ_{sp} matrix representation with generators $\frac{1}{2}\sigma^{\mu\nu}$. We check the result by confirming that the current of the transformed spinor is now proportional to the new momentum.

```
In[61]:= Asp = MatrixExp[- $\frac{i}{2}$  w[-μ, -ν] *  $\frac{1}{2}$  σ[μ, ν]] // GetArray // FullSimplify;
ψnew = Asp.ψ // FullSimplify;
bar[ψnew].γ[μ].ψnew // GetArray // FullSimplify
% == 2 pnew /. p0 →  $\sqrt{m^2 + p_1^2 + p_2^2 + p_3^2}$  // FullSimplify

Out[63]= { $\frac{2(\sqrt{m^2 + p_1^2 + p_2^2 + p_3^2} + v p_3)}{\sqrt{1-v^2}}$ , 2 p1, 2 p2,  $\frac{2(p_3 + v\sqrt{m^2 + p_1^2 + p_2^2 + p_3^2})}{\sqrt{1-v^2}}$ }

Out[64]= True
```

We can also check the transformation properties of the Dirac matrices directly $\Lambda_{\text{sp}}^{-1}\gamma^\mu\Lambda_{\text{sp}} = \Lambda_{\text{vec}}^\mu{}_\nu\gamma^\nu$.

```
In[65]:= GetArray[Inverse[Asp].γ[μ].Asp] == Avec.GetArray[γ[μ]] // FullSimplify

Out[65]= True
```

Lastly we clear the Minkowski space to not interfere with the next part.

```
In[66]:= DefineSpace["Minkowski", 0]
```

C. Chiral Kinetic Theory

In the last part we will examine applications of SimpleTensor to elementary differential geometric calculations on the example of chiral kinetic theory. It is based mostly on the excellent papers by Stephanov, Yin, and others [10, 11]. Note, that this material is more complicated and less widely known compared to previous sections. Nevertheless, physics aside it illustrates how SimpleTensor allows calculations to closely follow the natural language of differential geometry. First, we will write the Lagrangian 1-form for a chiral particle and check that it is Lorentz invariant. Then we will construct the pre-symplectic form and find the evolution vector field for the chiral particle. Finally, we will find the modification to the phase space volume and check the Liouville theorem for it.

In its simplest form chiral particles are living in the Euclidean phase space - a product of 3-dimensional spatial and momentum spaces, coupled to a 1-dimensional time line.

```
In[67]:= DefineSpace["Spatial", 3, index → {i, j, k, l}, coord → {x, y, z}]
DefineSpace["Momentum", 3, index → {a, b, c, d}, coord → {p1, p2, p3}]
DefineSpace["Time", 1, index → {τ}, coord → {t}]

metric[-i, -j] ≡ IdentityMatrix[3]
metric[-a, -b] ≡ IdentityMatrix[3]
```

Next we define in the spatial space a momentum vector \vec{p} and in the momentum space a Berry potential vector \vec{a}_p that was derived in the previous part.

```
In[72]:= p[i] ≡ {p1, p2, p3}
ap[a] ≡ { $\frac{p_2 + \frac{p_2 p_3}{\sqrt{p_1^2 + p_2^2 + p_3^2}}}{2(p_1^2 + p_2^2)}$ ,  $-\frac{p_1 + \frac{p_1 p_3}{\sqrt{p_1^2 + p_2^2 + p_3^2}}}{2(p_1^2 + p_2^2)}$ , 0}
```

Let us also introduce a uniform electric $\vec{E} = (0, 0, E_z)$ and magnetic $\vec{B} = (0, 0, B_z)$ fields both pointing in the z-direction. We therefore define the following electromagnetic potential:

$$\begin{aligned} \text{In[74]}:= & \varphi = \text{Ez z}; \\ & \text{A[i]} \equiv \left\{ -\frac{1}{2} \text{Bz y}, \frac{1}{2} \text{Bz x}, 0 \right\} \end{aligned}$$

The dispersion $\epsilon = |\vec{p}| - \vec{B} \cdot \vec{s}$ of a relativistic chiral particle gets modified in the presence of the magnetic field due to its interaction with spin.

$$\text{In[76]}:= \quad \epsilon = \sqrt{\text{p1}^2 + \text{p2}^2 + \text{p3}^2} - \frac{\hbar \text{Bz p3}}{2 (\text{p1}^2 + \text{p2}^2 + \text{p3}^2)};$$

Now we have all ingredients to define the Lagrangian 1-form L of a chiral particle. Without the \hbar correction this is exactly the usual classic-mechanical Lagrangian $L_{c1} = \epsilon - \varphi - (\vec{p} + \vec{A}) \cdot \vec{v}$.

$$\text{In[77]}:= \quad \text{L} = (\text{p}[-\text{i}] + \text{A}[-\text{i}]) \text{basis}[\text{i}] - (\epsilon - \varphi) \text{dt} - \hbar \text{ap}[-\text{a}] \text{basis}[\text{a}] // \text{GetArray};$$

Let us prove that this Lagrangian is indeed Lorentz invariant, at least up to the second order in \hbar . For simplicity we consider a boost in the z-direction $\vec{\beta} = (0, 0, \beta_3)$. In this case the electric and magnetic fields do not change and therefore the Lagrangian is invariant with respect to the transformation if its Lie derivative in the direction of the vector field that generates this transformation is zero $\mathcal{L}_\delta L \approx 0$. Under an infinitesimal Lorentz boost $\vec{\beta}$ the coordinates change by $\delta t = \vec{\beta} \cdot \vec{x}$, $\delta \vec{x} = t \vec{\beta}$ and $\delta p = \epsilon \vec{\beta}$. This results in the following vector field for a regular particle $\delta = (\vec{\beta} \cdot \vec{x}) \partial t + t (\vec{\beta} \cdot \partial \vec{x}) + \epsilon (\vec{\beta} \cdot \partial \vec{p})$. In the case of the chiral particle this field gets an additional so-called side-jump term, i.e., position of a chiral particle depends on the Lorentz frame.

$$\begin{aligned} \text{In[78]}:= & \text{With}[\{\beta = \{0, 0, \beta_3\}, \text{xvec} = \{x, y, z\}, \text{pvec} = \{p_1, p_2, p_3\}, \text{Bvec} = \{0, 0, \text{Bz}\}\}, \\ & \text{With}[\{\text{jump} = \frac{\hbar \beta \times \text{pvec}}{2 (\text{p1}^2 + \text{p2}^2 + \text{p3}^2)}\}, \\ & \quad \delta = \beta \cdot \text{xvec} \partial t + (t \beta + \text{jump}) \cdot \{\partial x, \partial y, \partial z\} + (\epsilon \beta + \text{jump} \times \text{Bvec}) \cdot \{\partial p_1, \partial p_2, \partial p_3\}; \\ & \quad \text{]]} \end{aligned}$$

Now we define the Lie derivative for differential forms using the Cartan formula $\mathcal{L}_X \omega = \iota_X d\omega + d(\iota_X \omega)$ and show that the Lagrangian is invariant (up to an irrelevant total derivative and up to \hbar^2) to the transformation generated by the boost δ .

$$\begin{aligned} \text{In[78]}:= & \text{LieDerivative}[\text{v}_-, \text{f}_-] := \text{InteriorProduct}[\text{v}, \text{ExteriorDerivative}[][\text{f}]] \\ & \quad + \text{ExteriorDerivative}[][\text{InteriorProduct}[\text{v}, \text{f}]] \\ & \text{LieDerivative}[\delta, \text{L}] - \text{ExteriorDerivative}[]\left[\frac{1}{2} \text{Ez} \beta_3 (t^2 + z^2) + \frac{\hbar \text{Bz} \beta_3 (\text{p1} x + \text{p2} y)}{4 (\text{p1}^2 + \text{p2}^2 + \text{p3}^2)}\right] + \text{O}[\hbar]^2 \\ \text{Out[79]}= & \quad \text{O}[\hbar]^2 \end{aligned}$$

The Lagrangian 1-form (without the dt term) works as a soldering form in our phase space and glues the particle velocity to its momentum. The exterior derivative of a soldering form gives the symplectic form of the corresponding phase space. However, our Lagrangian has an extra time coordinate and a term connected to it. The exterior derivative dL gives what is called a pre-symplectic form, i.e., an odd-dimensional 2-form whose null vector gives the system evolution and the quotient space with respect to this vector gives the phase-space. The projection of dL to this phase space is the symplectic form. Let us therefore calculate the exterior derivative of L .

$$\begin{aligned} \text{In[80]}:= & \text{dL} = \text{ExteriorDerivative}[][\text{L}] // \text{FullSimplify} \\ \text{Out[80]}= & \text{dp1} \wedge \text{dx} + \text{dp2} \wedge \text{dy} + \text{dp3} \wedge \text{dz} - \text{Ez} \text{dt} \wedge \text{dz} + \text{Bz} \text{dx} \wedge \text{dy} - \frac{\text{p1} \text{dp1} \wedge \text{dt} + \text{p2} \text{dp2} \wedge \text{dt} + \text{p3} \text{dp3} \wedge \text{dt}}{\sqrt{\text{p1}^2 + \text{p2}^2 + \text{p3}^2}} \\ & - \hbar \left[\frac{\text{p1} \text{dp2} \wedge \text{dp3} - \text{p2} \text{dp1} \wedge \text{dp3} + \text{p3} \text{dp1} \wedge \text{dp2}}{2 (\text{p1}^2 + \text{p2}^2 + \text{p3}^2)^{3/2}} + \frac{\text{Bz} \text{dp3} \wedge \text{dt}}{2 (\text{p1}^2 + \text{p2}^2 + \text{p3}^2)} \right. \\ & \quad \left. - \frac{\text{Bz} \text{p3} (\text{p1} \text{dp1} \wedge \text{dt} + \text{p2} \text{dp2} \wedge \text{dt} + \text{p3} \text{dp3} \wedge \text{dt})}{(\text{p1}^2 + \text{p2}^2 + \text{p3}^2)^2} \right] \end{aligned}$$

`ExteriorDerivative` with empty first brackets is just a shortcut to take the derivative in the product space of all defined spaces.

```
In[81]:= dL == ExteriorDerivative["Spatial", "Momentum", "Time"] [L]
        == ExteriorDerivative["Spatial"] [L] + ExteriorDerivative["Momentum"] [L]
        + ExteriorDerivative["Time"] [L] // FullSimplify

Out[81]= True
```

Let us now find the evolution equations. We define a general vector field $v = \partial t + \vec{v}_x \cdot \partial \vec{x} + \vec{v}_p \cdot \partial \vec{p}$ and then demand it to be the null vector of dL , i.e., its contraction with must be zero $\iota_v dL = 0$. The components of the form on the left-hand side are independent and therefore we get the system of evolution equations.

```
In[82]:= vspatial[i] == {vx, vy, vz}
vmomentum[a] == {vp1, vp2, vp3}
v = Dt + vspatial[i] basis[-i] + vmomentum[a] basis[-a] // GetArray
EqEvolution = CoefficientArrays[InteriorProduct[v, dL], {dt, dx, dy, dz, dp1, dp2, dp3}]
// Last // Normal // FullSimplify

Out[84]= Dt + Dp1 vp1 + Dp2 vp2 + Dp3 vp3 + Dx vx + Dy vy + Dz vz

Out[85]= - (p1 vp1 + p2 vp2 + p3 vp3) / Sqrt[p1^2 + p2^2 + p3^2] + Ez vz + (h Bz vp3) / (2 (p1^2 + p2^2 + p3^2)) - (h Bz p3 (p1 vp1 + p2 vp2 + p3 vp3)) / (p1^2 + p2^2 + p3^2)^2

vp1 - Bz vy
vp2 + Bz vx
vp3 - Ez

- vx + p1 / Sqrt[p1^2 + p2^2 + p3^2] + (h Bz p1 p3) / (p1^2 + p2^2 + p3^2)^2 + (h (p3 vp2 - p2 vp3)) / (2 (p1^2 + p2^2 + p3^2)^(3/2))
- vy + p2 / Sqrt[p1^2 + p2^2 + p3^2] + (h Bz p2 p3) / (p1^2 + p2^2 + p3^2)^2 + (h (p1 vp3 - p3 vp1)) / (2 (p1^2 + p2^2 + p3^2)^(3/2))
- vz + p3 / Sqrt[p1^2 + p2^2 + p3^2] - (h Bz (p1^2 + p2^2 - p3^2)) / (2 (p1^2 + p2^2 + p3^2)^2) - (h (p1 vp2 - p2 vp1)) / (2 (p1^2 + p2^2 + p3^2)^(3/2))
```

Note, that disregarding \hbar corrections the above equations amount to $\dot{x} = \vec{p}/|\vec{p}|$, i.e., a particle moving with a speed of light in the direction of its momentum, and $\dot{p} = \vec{E} + \vec{v} \times \vec{B}$, i.e., the Lorentz force. This system of equations indeed has the unique solution for the system evolution. The resulting vector v_{evol} , however, is quite cumbersome, its value can be found in the SimpleTensor example notebook.

```
In[86]:= vEvol = v /. First@Solve[EqEvolution == 0, {vx, vy, vz, vp1, vp2, vp3}] // Normal // FullSimplify;
```

Due to complexity of the evolution it is hard to actually find the canonical coordinates of the phase space and project dL onto them. It is still easy, however, to find the phase space volume, as it can be found from the highest non-zero power of the symplectic form. Note the \hbar correction in the brackets, its origin is intimately connected to the Berry potential \vec{a}_p and the chiral anomaly.

```
In[87]:= vol = 1/3! dL^dL^dL^dt // FullSimplify

Out[87]= (1 + (h Bz p3) / (2 (p1^2 + p2^2 + p3^2)^(3/2))) dp1^dp2^dp3^dt^dx^dy^dz
```

Let us finally check that the Liouville theorem in this theory still holds and the volume form stays invariant during the evolution. In other words, we show that its Lie derivative in the direction of evolution is zero.

```
In[88]:= LieDerivative[vEvol, vol] // FullSimplify
```

```
Out[88]= 0
```

Lastly, after all the calculations we tidy up after ourselves.

```
In[89]:= DefineSpace["Spatial", 0]
DefineSpace["Momentum", 0]
DefineSpace["Time", 0]
```

IV. SUMMARY

SimpleTensor is the new Mathematica package for componentwise tensor and differential-geometric calculations. Among its advantages (or at least distinctions) over alternatives are the ease of use, conciseness, flexibility, and hackability. It provides the most straightforward route to the result with minimal boilerplate code and is designed to be easily accessible. It contains only a handful of functions on top of which a user can build up depending on her or his needs, as we have done in the examples with `LieDerivative` or `HodgeStar`. Moreover, the codebase itself is short, documented, and supplemented by tests, which enables correction to the package itself, if need arises. Note, that due to deliberate lack of optimisation in favor of code clarity SimpleTensor is slower compared to the alternatives and is not suited for heavy calculations. SimpleTensor is also not specialized, providing only essential functionality, and so it lacks many frequently used functions for general relativity or quantum field theory. Despite these limitations, as illustrated in the examples, it can be used quite easily in settings ranging from an undergraduate level electrodynamics to relativistic quantum mechanics to even actively researched chiral kinetic theory.

-
- [1] <http://www.xact.es/>
 - [2] <http://www.stargazing.net/yizen/Tensoria.html>
 - [3] <http://www.vaudrevange.com/pascal/grt/>
 - [4] <https://sites.math.washington.edu/~lee/Ricci/>
 - [5] <https://github.com/drybalka/SimpleTensor>
 - [6] J. D. Jackson, *Classical Electrodynamics*, (Wiley, 1975)
 - [7] L. D. Landau and E. M. Lifshits, *The classical theory of fields*, (Butterworth-Heinemann, Oxford, 1979)
 - [8] M. E. Peskin and D. V. Schroeder, *An Introduction to quantum field theory*, (Addison-Wesley, Reading, USA, 1995)
 - [9] M. V. Berry, Proc. Roy. Soc. Lond. A **392**, 45-57 (1984)
 - [10] M. A. Stephanov and Y. Yin, Phys. Rev. Lett. **109**, 162001 (2012)
 - [11] J. Y. Chen, D. T. Son, M. A. Stephanov, H. U. Yee and Y. Yin, Phys. Rev. Lett. **113**, no.18, 182302 (2014)